



Weld your way.

## Carola - Požadavky

Požadavky na programový  
kód Carola

## Obsah

|   |          |
|---|----------|
| <b>1 Úvod</b>                                     | <b>3</b> |
| <b>2 Celková efektivita zařízení (OEE)</b>        | <b>3</b> |
| 2.1 Určení skutečného množství a dobrého množství | 3        |
| 2.1.1 Magisterský program                         | 3        |
| 2.1.2 Proměnné                                    | 6        |
| 2.1.3 Jednoduchý program komponent                | 6        |
| 2.1.4 Komplexní program složek                    | 7        |
| <b>3 Poznámky</b>                                 | <b>9</b> |
| <b>4 Proměnné</b>                                 | <b>9</b> |

## 1 Úvod

U některých funkcí C-Gate je nutné, aby byly informace z programu Carola předávány na C-Gate. Potřebné bloky a pokyny pro příslušnou funkci naleznete v příslušných kapitolách.

Pokud používáte datovou strukturu `ReadVar` nebo příkaz `OPCUAVAR` v jiném kontextu, ujistěte se, že se použité oblasti nepřekrývají. Upozorníme také, že index použitý v programu Carola není stejný jako index na webovém rozhraní. Hodnoty jsou předávány do C-Gate prostřednictvím datové struktury `ReadVar`. V kontroléru je první index 1 a poslední 100. V programu C-Gate je první index 0 a poslední 99. Hodnotu, která je v kontroléru uložena pod indexem 50, lze v programu C-Gate získat pod indexem 49.

Vezměte prosím na vědomí, že níže vysvětlené příklady kódu Carola mohou být již částečně nebo zcela uloženy v magisterském programu. Porovnejte prosím jednotlivé části a v případě potřeby je upravte. Duplikace funkcí a proměnných je interpretem Carola rozpoznána jako chyba.

**Poznámka:** Změny proměnných `QIROX WriteVar` mohou ovlivnit programovou sekvenci Carola.

**Poznámka:** Pokud chcete zkopírovat kód programu Carola, použijte k otevření tohoto souboru program Adobe Acrobat Reader.

## 2 Celková efektivita zařízení (OEE)

### 2.1 Určení skutečného množství a dobrého množství

Pro určení skutečného množství a dobrého množství musí být v hlavním programu a v programu komponent uloženy následující části programu. Důležité je, že následující postupy používají výhradně rozsah indexů 94 až 100 a nelze je použít pro jiné účely. Tento rozsah se může v budoucnu rozšířit.

#### 2.1.1 Magisterský program

První řádky jsou nutné k tomu, aby bylo možné proměnné importovat a zapisovat z jiných programů. Jsou umístěny bezprostředně za instrukcí `RESTART`.

```
RESTART
! DEKLARACE ĚPROMNNÉ
VAR SEL, ERG
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE, COMPVARI, COMPSERI, COMPSTEP, COMPSEAM
```

Tento blok definuje proceduru, která odesílá na adresu C-Gate informaci o spuštění komponenty. Předtím musí být v programu komponenty nastaveny proměnné použité níže.

```
PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
WAITM(100)
```

```

STRHND(2, COMPTYPE, COMPTYPE, ERG)
IF ERG=0 THEN COMPTYPE:='NODATA'
STRHND (0,COMPTYPE,'NODATA',ERG)
IF ERG=1 THEN BEGIN
LOPCUA1:
WRITE ('ĚPROMNÁ >COMPTYPE< DOSUD NEPOPSANÁ!')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99,COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98,COMPSEAM,'COMPSEAM')
OPCUAVAR(0,97,COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
ENDP

```

Tento blok definuje proceduru, která předává na adresu C-Gate informaci o tom, že komponenta je hotová.

```

! KONEC ZÁZNAMU ČSOUÁSTI
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')
WAITM(100)

COMPTYPE := ''
COMPVARI := ''
COMPSEAM := ''
COMPSTEP := ''
COMPSTAT := ''
COMPQUAL := ''
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99,COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98,COMPSEAM,'COMPSEAM')
OPCUAVAR(0,97,COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')

```

```
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Tento blok definuje funkci, která předává na adresu C-Gate informaci o tom, že se začíná vytvářet nový šev. Název švu musí být před voláním procedury zapsán do proměnné COMPSEAM.

```
! ZAHÁJENÍ ŠVU
PUBLIC PROC SEAMSTRT
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Tento blok definuje funkci, která předává na adresu C-Gate informaci o tom, že aktuální šev je dokončen.

```
! KONEC ŠVU
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
WAITM(100)
ENDP
```

Tento blok definuje funkci, která se na základě PHG dotazuje, zda je komponenta v pořádku, nebo ne. Kontrola se opírá o vizuální kontrolu pracovníkem a blokuje další tok programu, dokud není proveden zápis. Znění JE KOMPONENTA OK=1 NEBO NENÍ OK=0 můžete přizpůsobit svým požadavkům.

```
! ZÁZNAM KVALITNÍCH ČSOUÁSTÍ
PUBLIC PROC COMPQMAN
WREAD ('JE KOMPONENTA OK=1 NEBO NENÍ OK=0 ?', SEL )
IF SEL=1 THEN COMPQUAL := 'OK' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL, 'COMPQUALITY')
WAITM(100)
ENDP
```

Tento blok definuje funkci, která automaticky (bez dotazu) nastaví kvalitu komponenty na OK. Tuto funkci lze použít, pokud nechcete dynamicky zaznamenávat kvalitu celkové efektivity zařízení (OEE), ale přesto chcete mít celou metriku správně vypočítanou.

```
! ZÁZNAM KVALITNÍCH ČSOUÁSTÍ
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL, 'COMPQUALITY')
WAITM(100)
ENDP
```

## 2.1.2 Proměnné

V programu lze použít následující proměnné.

- **COMPTYPE** : Typ součásti.
- **COMPVARI** : Varianta komponenty.
- **COMPSEAM** : Název švu. Šev se může skládat z několika segmentů pásu.
- **COMPSTEP** : Název výrobního kroku. Hodnota je důležitá, pokud je součást zpracovávána na více stanicích.
- **COMPSEAM** : Název švu. Šev se může skládat z několika segmentů pásu.
- **COMPSERI** : Sériové číslo součásti. Pokud není nastavena žádná hodnota, C-Gate automaticky vytvoří sériové číslo na základě časového razítka.

## 2.1.3 Jednoduchý program komponent

Importujte proměnné a postupy z hlavního programu.

```
RESTART
```

```
! IMPORT ĚPROMNNÝCH A PROCEDUR  
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEAM,COMPSTEP,COMPSEAM FROM MASTER  
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
```

```
MAIN
```

V toku programu nastavte alespoň povinná pole a v případě potřeby také nepovinná pole. Jakmile jsou všechna pole nastavena, zavolejte proceduru **COMPSTART**. Pokud jste si zakoupili modul C-Gate.QM, můžete dodatečně definovat švy a strukturovat záznam součásti jemněji. K tomu použijte procedury **SEAMSTRT** a **SEAMEND**. Jakmile je díl dokončen a chcete nastavit kvalitu dílu z programu Carola, zavolejte buď **COMPQAUT**, nebo **COMPQMAN** pro přenos dat kvality. Pokud jste si zakoupili modul C-Gate.QM, kvalita dílu se automaticky stanoví pomocí sledování QIROX SD. Vyvolání procedury **COMPQAUT** nebo **COMPQMAN** nemá na protokol součásti žádný vliv.

Chcete-li komponentu dokončit, zavolejte proceduru **COMPEND**. Pořadí volání je důležité.

Upozorňujeme, že mezi voláním **COMPEND** a opětovným spuštěním komponenty pomocí **COMPSTART** musí uplynout až jedna sekunda, jinak může dojít ke ztrátě koncového signálu v důsledku synchronizace dat v řídicí jednotce. Plánuje se, že toto omezení bude odstraněno v některé z příštích verzí softwaru pro řídicí jednotku robota.

```
! NASTAVENÍ INFORMACÍ O ČSOUÁSTI  
! POVINNÉ POLE  
COMPTYPE := 'OVLÁDACÍ ŘŇSKÍ'  
  
! ĚVOLITELN  
! COMPVARI := 'XL'  
  
! ĚVOLITELN  
! COMPSEAM := '1234567890'
```

```

! ĚVOLITELN
! COMPSTEP := 'PODLAHA'

! ČZAÁTEK KOMPONENTY
CALL COMPSTART

! *****
! PROGRAM ŽSLOEK
! *****

COMPSEAM := 'ŘSVAENEC 1'
CALL SEAMSTR

! *****
! PROGRAM ŘSVAENEC 1. ŘNAPÍKLAD ŰRZNÁ HNUŤÍ GC.
! *****

COMPSEAM := 'ŘSVAOVACÍ SVÁR 2'
CALL SEAMSTR

! *****
! PROGRAM ŘSVAOVACÍ SVÁR 2. ŘNAPÍKLAD ŰRZNÁ HNUŤÍ GC.
! *****

CALL SEAMEND

! NASTAVENÍ KVALITY KOMPONENTY V PROGRAMU CAROLA
! ZAVOLEJTE ŘPED COMPEND
CALL COMPQMAN
! NEBO
! CALL COMPQAUT

! KONEC KOMPONENTY
! ZAVOLEJTE NA COMPQMAN NEBO COMPQAUT
CALL COMPEND

! ČUKONENÍ PROGRAMU ŽSLOKY
END

```

#### 2.1.4 Komplexní program složek

Následující program Carola vytvoří komponentu se dvěma výrobními kroky. Výrobní kroky lze svařovat buď postupně na jednom robotu, nebo paralelně na několika robotech. Předpokladem je, že jednotlivé kroky jsou protokolárně spojeny:

- Typ, varianta a sériové číslo musí být u všech robotů nastaveny stejně.

- Příslušná šablona komponenty musí mít nastaven správný minimální počet výrobních kroků. V tomto případě se svařují dva výrobní kroky, proto musí být počet nastaven alespoň na hodnotu dva.

```

RESTART
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
MAIN

! *****
! VÝROBNÍ KROK 1
! *****

COMPTYPE := 'KOTEL'
COMPVARI := '80KW'
COMPSEAM := '0042012342'
COMPSTEP := 'PODLAHA'

CALL COMPSTART

COMPSEAM := 'ŘSVAENEC 1'
CALL SEAMSTRT
! PROGRAM ŘSVAENEC 1. ŘNAPÍKLAD ÚRZNÁ HNUTÍ GC.

COMPSEAM := 'FILETOVÝ SVAR 2'
CALL SEAMSTRT
! PROGRAM FILETOVÝ SVAR 2. ŘNAPÍKLAD ÚRZNÁ HNUTÍ GC.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

! *****
! VÝROBNÍ KROK 2
! *****

COMPTYPE := 'KOTEL'
COMPVARI := '80KW'
COMPSEAM := '0042012342'
COMPSTEP := 'OBÁLKA'

CALL COMPSTART

COMPSEAM := 'ŘSVAOVACÍ SVÁR 3'
CALL SEAMSTRT
! PROGRAM ŘSVAOVACÍ SVÁR 3. ŘNAPÍKLAD ÚRZNÁ HNUTÍ GC.

```



```

COMPSEAM := 'ŘSVAOVACÍ SVÁR 4'
CALL SEAMSTRT
! PROGRAM ŘSVAOVACÍ SVÁR 4. ŘNAPÍKLAD ŮRZNÁ HNUTÍ GC.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

END

```

### 3 Poznámky

Pokud jste nebo jste byli zákazníkem PDM nebo UMS a v konfiguraci řídicí jednotky je aktivována jedna z těchto dvou možností, je třeba bezodkladně dodržet následující upozornění! Pokud je jedna z obou možností aktivována, ale řídicí jednotka robota není připojena k příslušné aplikaci na PC, budou v datovém toku vznikat mezery, které mohou být dlouhé až 5 sekund a výrazně falšovat protokol součástek. Buď připojte řídicí jednotku robota k aplikaci PC, nebo nechte příslušnou volbu z konfigurace odstranit.

Pokud pracujete s několika roboty současně na jedné komponentě a chcete mít všechna data spojena v jednom protokolu komponenty v rámci modulu C-Gate.QM, obraťte se na servisní službu CLOOS, kde získáte další informace. V opačném případě ponechte pole `COMPSEMI` prázdné, protože to může vést k nežádoucím vedlejším účinkům při výpočtu celkové efektivity zařízení (OEE).

### 4 Proměnné

V programu Carola máte možnost připojit k protokolu komponenty proměnné. K tomu slouží příkaz `OPCUAVAR` (viz příklad). Pokud má název (poslední parametr) předponu "COMPVAR\_", je proměnná připojena ke komponentě. Pokud má název předponu "STEPVAR\_", je proměnná připojena ke kroku součásti. Pokud má název předponu "SEAMVAR\_", je proměnná připojena k aktuálnímu švu. Pro každý vybraný název lze uložit pouze jednu hodnotu. V protokolu je uložena poslední hodnota, která byla naposledy nastavena s příslušným názvem. Předpona názvu se neukládá. Hodnotu je třeba vynulovat před dokončením švu, kroku nebo součásti. Pokud se tak nestane, je hodnota připojena i k následujícímu prvku. Lze ukládat celá čísla, čísla s čárkou a texty.

```

CALL SEAMSTRT
OPCUAVAR(0,1, 'EXAMPLE-VALUE', 'COMPVAR_SEAM_MEASURE_A')

OPCUAVAR(0,1, '', '')
CALL SEAMEND

```